Assoziativität des Wegprodukts

In[3]:=
```
$Assumptions = 0 ≤ t ≤ 1 && 0 ≤ s ≤ 1
```

Out[3]= $0 \le t \le 1 \;\&\&\; 0 \le s \le 1$

Wir wollen zeigen, dass das Wegprodukt assoziativ bis auf Homotopie relativ der Endpunkte ist. Dazu definieren wir zuerst das Produkt von Wegen:

In[4]:=
```
WegProdukt[α_, β_][t_] := Piecewise[
  {{α[2 t], 0 ≤ t ≤ 1/2},
   {β[2 t - 1], 1/2 ≤ t ≤ 1}}]
```

Berechnen wir damit die beiden möglichen Klammerungen. Ziel ist es, eine Homotopie zwischen den sich ergebenden Wegen zu finden.

```
WegProdukt[WegProdukt[α, β], γ][t] // Simplify
```

Out[6]=
$$\begin{cases} \alpha[4\,t] & t \le \frac{1}{4} \\ \beta[-1+4\,t] & \frac{1}{4} < t \le \frac{1}{2} \\ \gamma[-1+2\,t] & \text{True} \end{cases}$$

In[37]:=
```
WegProdukt[α, WegProdukt[β, γ]][t] // Simplify
```

Out[37]=
$$\begin{cases} \alpha[2\,t] & 2\,t \le 1 \\ \beta[-2+4\,t] & \frac{1}{2} < t \le \frac{3}{4} \\ \gamma[-3+4\,t] & \text{True} \end{cases}$$

Dazu unterteilen wir das Quadrat in 3 Bereiche, begrenzt durch 4 Geraden:

In[9]:=
```
separators[s_] := {0, 1/4 + 1/4 s, 1/2 + 1/4 s, 1}
```

Out[40]=



Die Idee ist, jedes der Vierecke seinerseits zu einem Quadrat deformieren und z.B. auf dem linken Viereck die "konstante" Homotopie zwischen α und sich selbst auszuführen.

In[24]:=
```
linel[s_] = separators[s][[2]];
liner[s_] = separators[s][[3]];
widthl[s_] = separators[s][[2]] - separators[s][[1]]
widthr[s_] = separators[s][[4]] - separators[s][[3]]
```

Out[26]= $\frac{1}{4} + \frac{s}{4}$

Out[27]= $\frac{1}{2} - \frac{s}{4}$

Definieren wir also die Deformationen ...

In[15]:=
```
Ass[α_, β_, γ_][t_, s_] :=
  {α[(t - 0)/widthl[s]],
   β[(t - linel[s])/(1/4)],
   γ[(t - liner[s])/widthr[s]]}
```

... und prüfen, dass sie auf den beiden Schrägen übereinstimmen:

In[30]:=
```
Ass[α, β, γ][linel[s], s][[{1, 2}]]
Ass[α, β, γ][liner[s], s][[{2, 3}]]
```

Out[30]= $\{\alpha[1], \beta[0]\}$

Out[31]= $\{\beta[1], \gamma[0]\}$

Denn nach Voraussetzung sind α und β sowie β und γ verknüpfbare Wege.
Damit können wir die Homotopie zwischen den verschiedenen Klammerungen definieren - diese ist stetig, da die Einschränkung auf jedes der abgeschlossenen Vierecke offensichtlich stetig ist.

In[16]:=
```
Assoziator[α_, β_, γ_][t_, s_] := Piecewise[
  {{Ass[α, β, γ][t, s][[1]], 0 ≤ t ≤ linel[s]},
   {Ass[α, β, γ][t, s][[2]], linel[s] ≤ t ≤ liner[s]},
   {Ass[α, β, γ][t, s][[3]], liner[s] ≤ t ≤ 1}}]
```

In[38]:=
```
H = Assoziator[α, β, γ];
H[t, s] // Simplify
```

Out[39]=
$$\begin{cases} \alpha\left[\frac{4t}{1+s}\right] & 1+s \geq 4t \\ \beta[-1-s+4t] & -2 \leq s-4t \leq -1 \\ \gamma\left[\frac{2+s-4t}{-2+s}\right] & 2+s \leq 4t \\ 0 & \text{True} \end{cases}$$

Wir testen, dass dies wirklich eine Homotopie zwischen $(\alpha\ \beta)\gamma$ und $\alpha\ (\beta\ \gamma)$ ist ...

In[20]:=
```
H[t, 0] // Simplify
H[t, 1] // Simplify
```

Out[20]=
$$\begin{cases} \alpha[4t] & 4t \leq 1 \\ \beta[-1+4t] & \frac{1}{4} < t \leq \frac{1}{2} \\ \gamma[-1+2t] & \text{True} \end{cases}$$

Out[21]=
$$\begin{cases} \alpha[2t] & 2t \leq 1 \\ \beta[-2+4t] & \frac{1}{2} < t \leq \frac{3}{4} \\ \gamma[-3+4t] & \text{True} \end{cases}$$

... die auch relativ der Endpunkte ist.

In[22]:=
```
Simplify[H[0, s]]
Simplify[H[1, s]]
```

Out[22]= $\alpha[0]$

Out[23]= $\gamma[1]$